財務計算あれこれ
第３回　内部利率の計算式
No. 3. Formulas for IRR and XIRR

(株) 竹内ハガネ商行　　竹内寿一郎

１．はじめに

もともと内部利率 IRR もしくはその拡張版である XIRR に興味を持ったのは志村氏の報告【1】、【2】からで、これまでの私の報告にもあるように、年価 (ローン返済額・均等積立額、または均等収益額) が入ってくると利率の計算は陽関数で求めることは出来ず、繰り返し演算例えば、前回述べた 2 分探索法による解法【6】とか、エクセルでよく使われるゴールシークなどの解法で解かなければならない。ところが J のパッケージ `intrest.ijs` の中にある関数 `IRR` および `XIRR` では何とニュートン法を採用し、しかもその表現が非常に簡便になされており【3】、これぞまさしくプロの作品であると感じたのでここに報告する次第である。

まず、`IRR` についてその計算式を述べ、どのように微分を求めてニュートン法を実現したかを解説し、それを拡張して毎年 (毎日) 返済額 (`IRR` では返済額は変わってもよいが、返済期日は一定間隔である) および返済間隔が変わるとしたときの関数 `XIRR` について、どのようにニュートン法を実現したかについて述べることにする。

２．内部利率・IRR の計算

内部利率計算式とは、期首にお金を借り、毎年または毎月定期的にある額 M_k をある期間返済してゆくときの借入額に対する金利を決めるための式である。この話は今 (期首に) 投資を行い、毎年または毎月ある額 M_k という利益が上げられるという場合の限界利率を算出するための計算式でもある。

返済パターンとして返済金 M_k を (期首, 期末) に支払うか否かによって (無, 無)、(無, 有)、(有, 無)、(有, 有) の 4 ケースが考えられるが【5】、期首に返済が有の場合は、借入金額から返済額を差し引いた金額が実際の期首の借入金額として、キャッシュフローを作成すればよい【6】。言うまでもないが (無, 有) がいわゆる期末払いであり、(有, 無) がいわゆる期首払いである。また簡単のため、ここでの説明は期末払いを例にとって説明する。

初期支払無、満期支払有 (期末)、　返済額は簡単のため一定の M とする



(1)　$$S = M + M(1+r) + \cdots + M(1+r)^{n-1} = M \times \frac{(1+r)^n - 1}{r}$$

(2)　$$P = M \times \frac{(1+r)^n - 1}{r(1+r)^n} = \frac{MS}{(1+r)^n}$$:年金現価係数 (期末)

(3)　$$M = P \times \frac{r(1+r)^n}{(1+r)^n - 1}$$:資本回収係数 (期末)

いま定期的に発生する金額 M を全て現在価値になおすことを考え、

$$(4)\quad \begin{cases} P &= \{M + M(1+r) + \cdots + M(1+r)^{n-1}\}/(1+r)^n \\ &= M\left\{\dfrac{1}{(1+r)^n} + \dfrac{1}{(1+r)^{n-1}} + \cdots + \dfrac{1}{(1+r)}\right\} \\ &= M\left\{x^n + x^{n-1} + \cdots + x^2 + x\right\} \qquad \text{ここで、}\quad x = \dfrac{1}{1+r} \quad \text{とした} \end{cases}$$

この式を x で微分すると 1 回微分関数が計算できる。

$$(5)\quad \begin{cases} P(x)' &= \;= \dfrac{dP(x)}{dx} = M\left\{nx^{n-1} + (n-1)x^{n-2} + \cdots + 2x + 1\right\} \\ &= \; M(n, (n-1), (n-2), \cdots, 3, 2, 1) \circledcirc (x^{n-1}, x^{n-2}, x^{n-3}, \cdots, x^2, x, 1) \\ & \qquad \text{上の式で、}\quad \circledcirc\text{はアダマールの積 (要素ごとの掛け算)} \end{cases}$$

もしも返済金額が時期に応じて変わる場合は、

$$(6)\quad P'(x) = (M_n, M_{n-1}, \cdots, M_2, M_1) \circledcirc (n, (n-1), \cdots, 2, 1) \circledcirc (x^{n-1}, x^{n-2}, \cdots, x^2, x, 1)$$

さてここで、ニュートン法について復習してみる。

3．ニュートン法

ニュートン法は陰関数における $F(x_0) = 0$ を満たす x_0 を繰り返し法で求めるための標準的手法で、その収束の早さは x_0 の近傍で 2 乗の早さであることが知られている。すなわち、解の精度が今 0.1 であったとすると次回は 0.01、その次は 0.0001、その次は 0.00000001 という具合に 2 乗のオーダーで真の解に収束してゆくと言われている。

いま $F(x)$ を解 x_0 の近傍でテーラー展開すると、

$$(6)\quad F(x_0 + \delta x) = F(x_0) + F'(x_0)\frac{\delta x}{1!} + O(\delta x^2) \qquad \delta x^2\text{以上のオーダーはゼロと見做して}$$

これを変形して、

$$(7)\quad \delta x = \frac{F(x_0 + \delta x) - F(x_0)}{F'(x_0)} = \frac{F(x_0 + \delta x)}{f(x_0)} = \frac{\text{今回の計算値}}{\text{解での微分値}} = \text{次回の修正項}$$

$F(x)$ を十分なめらかな増加関数とすると、$f(x)$ は正の連続関数で、前回の δx が正であれば $F(x_0 + \delta x)$ は正であるから、根はより小さいところにあるので次回の修正項は負にならなければならない。従って次回の繰り返しにおける x_{n+1} は、

$$(8)\quad x_{n+1} = x_n - \delta x$$

となる。

また、$F(x)$ がなめらかな減少関数とすると、$f(x)$ は負の連続関数で、前回の δx が正であれば $F(x_0 + \delta x)$ は負であり、$f(x_0)$ も負であるから、次回の修正項 δx は正、一方、根は今の値より小さいところにあるので、やはり $x_{n+1} = x_n - \delta x$ となるから新しい解は修正項を引くことによって得られる。

4．J による内部利率 irr の計算

J では実に簡単に関数が作成されている。

```
irr=: 3 : 0
0 irr y  NB. [1]
:
t=. %>:x  NB. [2]
```

```
cf=. ,y  NB. [3]
tol=. 1e_5  NB. [4]
max=. 15  NB. [5]
if. 2>#cf do. 'cash flow must have at least two elements' return. end.  NB. [6]
if. -. *./_1 1 e.*cf do. 'no sign change in cash flow' return. end.  NB. [7]
cf=. (|.cf),:|.1|.cf*i.#cf  NB. [8]
while. tol<|t-r=. t-%/t#.cf do. t=. r  NB. [9]
  if. 0=max=. <:max do. 'iterations exceeded' return. end.  NB. [10]
end.
<:%r  NB. [11]
)
```

上の J の関数を以下に詳しく説明する。関数における文の最後尾の [ ] が対応する説明である。

[1] 単項の場合、初期値はゼロとして繰り返し演算を行う

[2] 現在価値に直す基本の利率を $t = \dfrac{1}{(1+x)}$ とおく

[3] 右引数であるキャッシュフローをリスト化する (アトムを 1 と数えるため)

[4] 求める解の精度を 10^{-5} とする

[5] 繰り返し数の最大は 15 とする (実際には 4〜5 回で十分)

[6] キャッシュフローは 2 個以上の数値がないとダメ

[7] キャシュフローには少なくとも 1 つ以上の正と負の数値が混在しなければならない

[8] ここがこの関数の一番のキーポイント
`cf*i.#cf` は (5) 式のアダマール積を計算。`1|.` で 1 つ左へ回転 (シフト)
その後、逆順`|.` にする。そうすると先頭はゼロになっている。
`(|.cf),:`はそれと元のキャッシュフローとリスト同士で 2 行のテーブルとして連結
新しい `cf` は上が元のキャッシュフロー、下がその微分の多項式の係数になっている

[9] 古い利率を `t` として、`t#.cf` は上が多項式 $F(x_0+\delta x)$ の計算値、下がその微分値 $f(x_0+\delta x)$ の計算をして、それを`%/`で上の値を下で割り δx を計算している。
新しい利率 `r` は旧の `t` から修正項を引いて求め、新旧の差の絶対値が許容誤差以下であれば終わり

[10] 若しくは繰り返し数 `max` を 15 から 1 つずつ減らして、0 になったら終了

[11] (4) 式から求める利率 i は、$i = \dfrac{1}{r} - 1$ で計算される

[8] と [9] が非常に巧妙である。例として、印刷の関係で短くしたキャッシュフロー、`cf0=.(-50),(5#12)` としてみてみよう。そうすると、

```
   cf0
_50 12 12 12 12 12
   cf
12 12 12 12 12 _50
 0 60 48 36 24  12
```

となり上が逆順のキャッシュフロー。下が 1 つずらして自然数倍し逆順にしたフローで、実質、上段の各項の x の n-1 乗の微分係数 M × k になっている。

```
   1.05 #.cf    NB. 初期値を x=1.05 とする
19.623 205.386          NB. Base を使って cf を係数とした多項式の計算
   (|."1 cf) p. 1.05    NB. 多項式の関数を使って計算してもよい
19.623 205.386
```

#. も p. も引数を少し工夫しなければならないが、いずれも多項式の計算である

```
   print=:(1!:2)&2   NB. 外部関数を使って画面上に表示する
   1.05-print %/1.05 #.cf    NB. x=1.05 としてみる
0.0955417                     NB. 修正項をプリントしてみた
0.954458                      NB. 修正後の新しい x の値
   0.954458-print %/0.954458 #.cf    NB. 新しい値で計算してみた
0.0143408                            NB. 新しい値での修正項
0.940117                             NB. 次の新しい値
   0.940117-print %/0.940117 #.cf    NB. 求めた新々値で計算してみた
0.000287056                          NB. 新々値での修正項、前の 2 乗のオーダー
0.93983                              NB. 新々々値
   0.93983-print %/0.93983 #.cf   NB. 新々々値で計算してみた
1.68582e_7                        NB. 修正項はさらに 2 乗のオーダーで小さくなる
0.93983                 NB. これが求める x の値
   (%0.93983)-1         NB. x から利率 i を求める
0.0640222
```

5．拡張された内部利率・XIRR の計算

返済間隔が不定期のときの内部利率の計算で、次のキャッシュフローの式を満たす。
(ここでは期末で説明する。M_0 がある場合は 0 期を $-P + M_0$ にすればよい。)



$$(9) \quad F(r) = -P + \frac{M_1}{(1+r)^{t_1-t_0}} + \frac{M_2}{(1+r)^{t_2-t_0}} + \cdots + \frac{M_n}{(1+r)^{t_n-t_0}} = 0$$

ここで、$x = \dfrac{1}{(1+r)}$ とおくと、

(10) $\quad F(x) = -P + \sum_{k=1}^{n} x^{t_k - t_0} M_k$

であるから、これを x で微分して、

(11) $\quad \dfrac{dF(x)}{dx} = F'(x) = f(x) = \sum_{k=1}^{n} (t_k - t_0) M_k x^{t_k - t_0 - 1}$

微分は (11) 式により計算することになる。

6．J による拡張した内部利率・XIRR の計算

J による xirr は irr と同様にニュートン法で求めている。
例によって説明のために文の最後尾に [ ] をつけておく。

```
NB. example:
NB.
NB.    time=. 0 1.2 4.7 5
NB.    pay=. _100 10 8 120
NB.    0 xirr time,:pay
NB. 0.0713808
NB.
xirr=: 3 : 0
0 xirr y   NB. [1]
:
t=. %>:x   NB. [2]
'time cf'=. |."1 y   NB. [3]
tol=. 1e_5   NB. [4]
max=. 15   NB. [5]
if. 2>#cf do. 'cash flow must have at least two elements' return. end. NB.[6]
if. -. *./_1 1 e.*cf do. 'no sign change in cash flow' return. end. NB. [7]
df=. cf * time   NB. [8]
mp=. +/ . *   NB. [9]
while.
  r=. t - (cf mp t ^ time) % df mp t ^ time - 1   NB. [10]
  tol < |t-r do.   NB. [11]
  t=. r   NB. [12]
  if. 0=max=. <:max do. 'iterations exceeded' return. end.   NB. [13]
end.
<:%r   NB. [14]
)
```

J の関数の説明

[1] 単項の場合、初期値はゼロとして繰り返し演算を行う

[2] 現在価値に直す基本の利率を $t = \dfrac{1}{(1+x)}$ とおく

[3] 固定時間とキャッシュフローを逆順にして読み込み、`time` と `cf` とする
参考： `time ,: pay` と `> time ; pay` は同じものをつくる

[4] 求める解の精度を 10^{-5} とする

[5] 繰り返し数の最大は 15 とする (実際には 4〜5 回で十分)

[6] キャッシュフローは 2 個以上の数値がないとダメ

[7] キャシュフローには少なくとも 1 つ以上の正と負の数値が混在しなければならない

[8] 微分したときの $kM_k t^{k-1}$ におけるはじめの k と M_k の積の部分のリスト
t^{k-1} の部分は後で計算する

[9] 内積を定義する。行列積 (matrix product) にも使える

[10] `df mp t ^ time - 1` で $kM_k t^{k-1}$ の和を計算、つまり微分値の計算
(`cf mp t ^ time`) で各出入りの金額を全て現価に直して $F(x)$ を計算、これを微分で割って修正項を計算し、前回の `t` を修正する

[11] 前回の t と今回修正して得た r の絶対値を比較して、それが未だ誤差の許容範囲より大であればこれを繰り返す

[12] 誤差がまだ大きいときは t を新しい値 r に置き換えて繰り返しを行う

[13] 無限ループを避けるため、繰り返し数 `max` を 15 から 1 つずつ減らして、0 になったら終了

[14] (10) 式から分かるように求める利率 r は、$r = \dfrac{1}{x} - 1$ で計算される

Ｊの関数をみてわかるように `xirr` の方がニュートン法を使っているなぁという匂いがプンプンするが、先の `irr` はあまりにも巧妙すぎて、はじめはこれがニュートン法であるとはとても思えなかった。

７．Ｊによる irr および xirr 関数の例題

● 前回の復習、返済額が同じ場合を考え、既知の M に対する利率を求める
　ちょうど利率が 7%になるような M を用いて計算した。初期値は 0.05

```
   cf0=.(-100),(10#14.2378)    NB. (無,有)：期末
   0.05 irr cf0
0.0700008
   cf1=.(13.3063-100),(9#13.3063)    NB. (有,無)：期首
```

```
   0.05 irr cf1
0.0699998
   cf00=.(-100),(9#15.3486)   NB. (無,無)
   0.05 irr cf00
0.0699993
   cf11=.(12.4633-100),(10#12.4633)   NB. (有,有)
   0.05 irr cf11
0.0700008
```

● 同じ問題を IRR を使わずにエクセルのゴールシークで解く



	A	B	C	D	E	F	G
1	期	期末	利率	期首	無･無	利率	有･有
2			1.070000704	-100		1.0700007	-100
3	0	-100	1	-86.6937	-100	1	-87.5367
4	1	14.2378	0.934578824	13.3063	15.3486	0.93457883	12.4633
5	2	14.2378	0.873437579	13.3063	15.3486	0.87343759	12.4633
6	3	14.2378	0.816296266	13.3063	15.3486	0.81629628	12.4633
7	4	14.2378	0.762893204	13.3063	15.3486	0.76289322	12.4633
8	5	14.2378	0.712983834	13.3063	15.3486	0.71298385	12.4633
9	6	14.2378	0.666339594	13.3063	15.3486	0.66633961	12.4633
10	7	14.2378	0.622746874	13.3063	15.3486	0.62274689	12.4633
11	8	14.2378	0.582006041	13.3063	15.3486	0.58200606	12.4633
12	9	14.2378	0.543930522	13.3063	15.3486	0.54393054	12.4633
13	10	14.2378	0.508345948			0.50834597	12.4633
14	IRR	7.00007556%	=1/1.05^i	6.99998234%	6.99992800%	=1/1.05^i	7.00007508%
15							
16		=Bi*Ci		=Di*Ci	=Ei*Fi		=Gi*Fi
17	1	13.30634639		12.4357862	14.3444766		11.64793631
18	2	12.43582956		11.6222225	13.4060441		10.88591468
19	3	11.62226297		10.861883	12.529005		10.17374538
20	4	10.86192087		10.1512858	11.7093428		9.508167044
21	5	10.15132123		9.48717679	10.9433039		8.886131617
22	6	9.487209866		8.86651453	10.2273802		8.304790477
23	7	8.866545443		8.28645673	9.55829297		7.761481355
24	8	8.286485617		7.74434699	8.93297824		7.253716153
25	9	7.744373986		7.23770271	8.34857234		6.779169545
26	10	7.237727936					6.335668332
27	Sum	100.0000239		86.6933753	99.9993963		87.5367209

エクセルのゴールシークによる内部利率 IRR の計算

● intrest.ijs の中の xirr 関数の例題を、print 文を付加して実行してみる

```
    time=. 0 1.2 4.7 5
    pay=. _100 10 8 120
    0 xirr time,:pay
1                    NB. 初期値  t、利率を 0 としたときの t
600 37.6 12 0        NB. df=(|.cf)*time キャッシュフローの逆順の時間倍
0.0584975            NB. 最初の修正項
0.941502             NB. 最初の t
0.00799209           NB. 2 回目の修正項
0.93351              NB. 2 回目の t
```

```
0.000135423          NB. 3回目の修正項、前回のほぼ2乗になっている
0.933375             NB. 3回目のt
3.82448e_8           NB. 4回目の修正項、前回のほぼ2乗になっている
0.933375             NB. 4回目のt
0.0713808            NB. (%0.933375)-1=0.0713807
```

● 同じ問題をエクセルの XIRR 関数で解く

C2 =-100

	A	B	C	D	E	F	G
1		期	金額	日付	日にち間隔	J例題	
2		0	-100	2010/3/27	0	0	
3		1	10	2011/6/8	438	1.2	
4		2	8	2014/12/6	1715.5	4.7	
5		3	120	2015/3/26	1825	5	
6							
7		0.071382					
8		=XIRR(C2:C5,D2:D5,0.05)					
9							
10							
11							

エクセルの拡張された内部利率関数 XIRR で計算

● 今度はエクセルの財務関数である XIRR 関数の中の例題を J で解いてみよう。
はじめにエクセルの例題から



	A	B	C	D	E	F
1		キャッシュフロー	日付	経過日にち	年換算	
2	0	-10,000	2008年1月1日	39448	0	
3	1	2,750	2008年3月1日	60	0.164384	
4	2	4,250	2008年10月30日	303	0.830137	
5	3	3,250	2009年2月15日	411	1.126027	
6	4	2,750	2009年4月1日	456	1.249315	
7						
8		0.373362535				
9		=XIRR(B2:B6,C2:C6,0.1)				
10						
11						
12						
13						

エクセルの拡張された内部利率関数 XIRR の例題から

エクセルでは期間に日付を採用している。J では 1 年間を 365 日として、小数点を使った表現を使って、期日は年単位に直さなければならない。一方、エクセルでは日付を採用しているので、J の年単位に 365 をかけて基準日からの日付に変更しなければならない (日付の代わりに基準日からの日にち間隔で表してもよい)。従って小数点以下を四捨五入したりする関係で、精度の上からは必ずしも完全一致とはならないことに注意しよう。なお、エクセルでは日付の代わりに日にち間隔を用いても同じ結果を得ることができる。

次に J で同じ問題を解いてみる。

```
   pay=._10000 2750 4250 3250 2750
   pay
_10000 2750 4250 3250 2750
   time=.0 0.164383562 0.830136986 1.126027397 1.249315068
   time
0 0.164383562 0.830136986 1.1260274 1.24931507
   0 xirr time,:pay
0.373362534
```

エクセルでは 0.373362535 であったが J では 0.373362534 という結果を得た。
期日の計算でかなり多くの桁を使用したので最終桁の 1 つ前まで一致したが、実際にはもっと精度が落ちると考えられる。実際問題では土日や休祭日、営業日などの関係で日数が変わったり 1 年が 365 日でいいか、など問題が含まれている。

【参考文献】

【1】志村正人（2006）：　内部収益率・IRR、　JAPLA 研究会 2006.1.25 資料

【2】志村正人 (2009)：内部利子率と幾何平均収益率、　JAPLA2009 シンポジウム 2009.12.8 資料

【3】J601 システムパッケージ (1994-2006)　：ファイナンスパッケージ、j601/system/packages/finance/interest.ijs 、j601_win.exe を解凍すると得られる

【4】Office 2003 Microsoft Excel（2003）：　HELP　財務計算、Excel 内部関数資料

【5】竹内寿一郎（2010）：　財務計算あれこれ　第 1 回 __現価、終価、年価__、 JAPLA 研究会 2010.1.23 資料

【6】竹内寿一郎（2010）：　財務計算あれこれ　第 2 回 __期間と利率の計算式__、 JAPLA 研究会 2010.2.27 資料